General Relativity Tutorials

1.1 Manifolds

David Park, djmp@earthlink.net
http://home.earthlink.net/~djmp/

Initialization

The ExtendRotations package may be obtained from my web site above.

```
In[1]:= Needs["TensorCalculus4`Tensorial`"]
        Needs["DrawGraphics`DrawingMaster`"]
        Needs["Cardano3`ComplexGraphics`"]
        Needs["Rotations`ExtendRotations`"]
```

```
In[5]:= DeclareBaseIndices[{1, 2}]
        DefineTensorShortcuts[{{x}, 1}]
        DeclareIndexFlavor[{red, Red}]
        labs = {x, δ, g, Γ};
```

Routines

1. Definition of a Manifold

An n-dimensional **manifold** is a topological space that locally looks like $\mathbb{R}^n$. This means that we can make a set of invertible maps between open sets in the manifold and open sets in $\mathbb{R}^n$, such that each point in the manifold is in one of the maps and there are smooth transistion functions between the maps wherever they overlap.

In general relativity we will be dealing with 4-dimensional manifolds that are not embedded in any higher dimensional space. For purposes of illustration and obtaining experience we can deal with 2-dimensional manifolds, which we can visualize as embedded in 3-dimensional space.

2. Manifold Examples

■ 2.1 Example - Charts on a Physical Manifold

For our first case we consider a manifold that is a curved surface depicted in 3-dimensional space. To mathematically find our way around the manifold we need to put coordinates on it so we can identify positions. Since we say that a 2-dimensional manifold is everywhere locally like a little open piece of $\mathbb{R}^2$, we can envision mapping pieces of the manifold to $\mathbb{R}^2$. We might not always be able to cover the entire manifold with a single piece. The following graphic shows one **chart** that covers an open region of the manifold. (Evaluated the thin closed cell for the graphic.)

A Chart on Manifold M



Chart **Manifold M with Coordinate Patch**

The beige colored surface is our manifold, **M**. (We can consider it to be an open set of points without boundary.) Part of the manifold is covered by a **coordinate patch**. We can think of the coordinate patch as being the *open* set of points in **M** that is covered by the gray area. We have a **coordinate mapping function**, ψ, which maps the points in the coordinate patch to an *open* set in $\mathbb{R}^2$. The coordinate patch with the coordinate mapping function is called a **chart**. I will also use the term 'chart' to refer to the image of the coordinate patch. So the patch is on the manifold, the chart is in flat space $\mathbb{R}^n$, and ψ and ψ^{-1} get us between them. For a point **P** in the manifold, ψ returns a list of two real numbers...

```
In[13]:=  ψ[P] == {xu[1][P], xu[2][P]}
```

$$\text{Out[13]=}\quad \psi[P] == \{x^1[P],\ x^2[P]\}$$

The x^i are called the **coordinate functions** of the coordinate map. In this case, if we use **x** and **y** as the coordinate names, the chart is the open region $-10 < x < 10$ and $-10 < y < 10$. We insist that the coordinate map, ψ, be continuous, 1-1 and onto the chart. This means that ψ is invertible and $\psi^{-1}[x, y]$ will give us a point in the manifold, which we could label with the coordinates **x** and **y**. Thus, by moving a point around in the chart we can navigate around the actual manifold. Clearly, the coordinate patch, the mapping function and the chart all go together. Given two of them, we could in principle find the third one. The terminology in the literature is various and not always too clear. The important point is that we have a continuous 1-1 method of getting back and forth between the chart and associated coordinate patch on the manifold. Since the charts are in ordinary Cartesian space, we know how to do calculus on them. Our charts will allow us to extend the concepts of calculus to manifolds. I will say more about the nature of these charts below.

If the manifold has features on it, then our mapping function would transfer those features to the chart, and it becomes what in ordinary parlance is called a 'map'. But mathematicians usually use the term 'map' for the function that does the mapping.

In the following graphic we use a second chart to cover an additional region of the manifold.



We could continue adding charts to cover the entire manifold, and we could do this in many ways. In fact, one of the essential characteristics of a manifold is that it can be completely covered in this manner. The two coordinate patches also overlap and since on part of the manifold they specify the same points, there must be a relation between their coordinates.



Mathematicians say that there must be smooth maps between the two coordinate systems in the open region that constitutes their overlap. Mathematicians usually give examples where the manifold is embedded in a higher dimensional space and precisely defined in a mathematical sense. They then derive the maps that connect the two coordinate systems. We will see such a case as the next example. We can write our transfer functions as...

```
In[19]:= xu[red@a][xu[b]]
         % // EinsteinArgument[x]
```

Out[19]= $x^{a}[x^{b}]$

Out[20]= $x^{a}[x^{1}, x^{2}]$

and the transfer functions going the other way as...

```
In[21]:= xu[a][xu[red@b]]
         % // EinsteinArgument[x]
```

Out[21]= $x^a[x^b]$

Out[22]= $x^a[x^1, x^2]$

Here the different index colors refer to different charts. It will turn out later, that what we are interested in is the partial derivatives of one set of coordinates with respect to another set of coordinates.We saw in the Introduction notebook that there were many ways to calculate this in *Tensorial*. There will be more specific examples in succeeding notebooks.

As physicists or engineers, what are we to make of the manifold and the coordinate charts? The surface might be part of a surface of an automobile. There are no numbers or coordinates on the surface to begin with. Or, for us, the manifold might be a region of spacetime. But nature does not have any built-in numbers, other than those scattered around as human artifacts. If the manifold isn't a mathematician's creation, but some real physical thing, how do we connect it with the mathematics?

The only way we can get coordinates onto our physical manifold is to build physical devices that generate or create coordinates. This will involve a minor, or even a vast engineering project. For example, to put coordinates on your location as you move around the surface of the earth, you can buy a GPS (Global Positioning System) receiver at almost any electronics store. GPS is an elaborate system made up of satellites, atomic clocks, radio transmitters and receivers. It assigns a latitude and longitude to every point, which you can read out from the GPS receiver as two real numbers. And pairs of real numbers can be put into correspondence with $\mathbb{R}^2$ and that's mathematics! We could imagine there being several GPS systems, which for some quirky reason used different coordinate systems, and which perhaps worked over different patches of the earth. If you and a friend each had different GPS receivers giving different coordinate systems, you could walk around side by side and record the two sets of readings. You could then try to find a smooth mathematical function that converted between the two coordinate systems, perhaps using a routine like **`FindFit`** in *Mathematica*.

Or, for smaller surfaces, we might think of being able to buy small grids printed on some transparent stretchable material. We could stretch the grid patches taut over parts of the surface without tearing or folding. Thus each grid would provide coordinates for that region. In a region of overlap, we could take many data points by reading off cooresponding coordinates from the two grids. (We don't need numbers for the actual points on the manifold. We only need to know that two pairs of grid values coorespond to the same point.) Then, again, we could try to find a smooth transistion function between the two sets of coordinates.

If we have a mathematical model for a system, then we might derive coordinates from the particular model we are using and compare the results using the model with actual physical measurements.

Clearly, coordinates are something we *arbitrarily* impose on a surface or manifold and the actual physics should not depend upon our choice of coordinates. The physical entities exist on the manifold, but without numbers. Like generals fighting battles on military maps, while the real battle wages elsewhere, we do our calculus on the charts while the physical world exists on the real manifolds.

■ 2.2 Example - An Atlas for a Sphere

To make an atlas that covers a sphere we can use two stereographic projections, one from the north pole to the plane of the equator and one from the south pole to the plane of the equator. (There are many other ways to build an atlas for the sphere but for this example we will stay with variations of stereographic projections.) The following diagram illustrates the case for projection from the north pole.



This will cover every point on the sphere except the north pole. Let **{X, Y, Z}** be the coordinates of point **P** and **{x, y, 0}** be the coordinates of point **Q**. Let $R == \sqrt{X^2 + Y^2}$ and $r == \sqrt{x^2 + y^2}$. The following equations from the diagram allow us to solve for λ. We will say that the distance **NQ** is λ times the distance **NP** and solve for λ. It turns out to be convenient to express the first equation in complex form.

```
In[24]:= ptN = a ⅈ;
         ptP = a ⅇ^(ⅈ θ);
         eqns = {ComplexExpand[Im[ ptN + λ (ptP - ptN)]] == 0, R^2 + Z^2 == a^2, r == λ R, Sin[θ] == Z / a}
         λsol = Part[Solve[eqns, λ, {R, Z, Sin[θ]}], 1, 1]
```

Out[26]= $\left\{a - a\lambda + a\lambda\,\text{Sin}[\theta] == 0,\ R^2 + Z^2 == a^2,\ r == R\lambda,\ \text{Sin}[\theta] == \frac{Z}{a}\right\}$

Out[27]= $\lambda \to \frac{a^2 + r^2}{2\,a^2}$

Solving for **P**...

```
In[28]:= Solve[Q == N + λ (P - N), P][[1,1]] // FullSimplify
```

Out[28]= $P \to \frac{Q + N\,(-1 + \lambda)}{\lambda}$

Writing this as a vector equation and solving...

```
In[29]:= {X, Y, Z} == ({x, y, 0} + {0, 0, a} (λ - 1)) / λ
         step1 = % /. λsol // Simplify
         Print["Cartesian coordinate form of mapping"]
         cartesianStereographic = step1 /. r^2 → x^2 + y^2
         Print["Polar form of solution"]
         polarStereographic = step1 /. {x → r Cos[ϕ], y → r Sin[ϕ]}
```

Out[29]= $\{X, Y, Z\} == \left\{\frac{x}{\lambda}, \frac{y}{\lambda}, \frac{a(-1+\lambda)}{\lambda}\right\}$

Out[30]= $\{X, Y, Z\} == \left\{\frac{2 a^2 x}{a^2 + r^2}, \frac{2 a^2 y}{a^2 + r^2}, \frac{-a^3 + a r^2}{a^2 + r^2}\right\}$

```
Cartesian coordinate form of mapping
```

Out[32]= $\{X, Y, Z\} == \left\{\frac{2 a^2 x}{a^2 + x^2 + y^2}, \frac{2 a^2 y}{a^2 + x^2 + y^2}, \frac{-a^3 + a (x^2 + y^2)}{a^2 + x^2 + y^2}\right\}$

```
Polar form of solution
```

Out[34]= $\{X, Y, Z\} == \left\{\frac{2 a^2 r \operatorname{Cos}[\phi]}{a^2 + r^2}, \frac{2 a^2 r \operatorname{Sin}[\phi]}{a^2 + r^2}, \frac{-a^3 + a r^2}{a^2 + r^2}\right\}$

The Cartesian solution uses **{x, y}** as our coordinates and the polar solution uses **{r, ϕ}** as our coordinates. The polar form will be more convenient for graphics but contains a missing line at $\phi == 0$. The projection from the south pole is just a reflection in the equatorial plane and hence only reverses the sign of the **Z** component.

The following graphic show us the north and south pole stereographic projections of a region in the plane $0 < r < 2$ and $0 < \phi < 2\pi$. We could increase the radius **r** and cover more of the sphere with each chart. As you can see, the constant **r** lines become more closely spaced as we approach the projection pole. As long as each chart covers more than a hemisphere, the two charts overlap and together cover the entire sphere.

Two Coordinate Patches Covering the Sphere
Polar Coordinates



North Pole Projection **South Pole Projection**

The following shows the coordinate patches for a rectangular region $-2 < x < 2$ and $-2 < y < 2$. (Stereographic projections of rectangular regions alway gives us a serrated edge of four circular arcs,)

Two Coordinate Patches Covering the Sphere
Cartesian Coordinates



North Pole Projection **South Pole Projection**

The functions that connect the different coordinate patches are called **transistion functions** (by Penrose). Let `{x, y}` denote the Cartesian coordinates for a north pole projection and use `{u, v}` for the Cartesian coordinates in the south pole projection. Then we can easily solve for one set in terms of the other.

```
In[37]:= Part[cartesianStereographic, 2] ==
          (Part[cartesianStereographic, 2] {1, 1, -1} /. {x → u, y → v})
         First@Solve[%, {x, y}]
```

Out[37]= $$\left\{\frac{2 a^2 x}{a^2+x^2+y^2}, \frac{2 a^2 y}{a^2+x^2+y^2}, \frac{-a^3+a\,(x^2+y^2)}{a^2+x^2+y^2}\right\} == \left\{\frac{2 a^2 u}{a^2+u^2+v^2}, \frac{2 a^2 v}{a^2+u^2+v^2}, -\frac{-a^3+a\,(u^2+v^2)}{a^2+u^2+v^2}\right\}$$

Out[38]= $$\left\{x \to \frac{a^2 u}{u^2+v^2}, y \to \frac{a^2 v}{u^2+v^2}\right\}$$

```
In[39]:= Part[cartesianStereographic, 2] ==
          (Part[cartesianStereographic, 2] {1, 1, -1} /. {x → u, y → v})
         solsouth = First@Solve[%, {u, v}]
```

Out[39]= $$\left\{\frac{2 a^2 x}{a^2+x^2+y^2}, \frac{2 a^2 y}{a^2+x^2+y^2}, \frac{-a^3+a\,(x^2+y^2)}{a^2+x^2+y^2}\right\} == \left\{\frac{2 a^2 u}{a^2+u^2+v^2}, \frac{2 a^2 v}{a^2+u^2+v^2}, -\frac{-a^3+a\,(u^2+v^2)}{a^2+u^2+v^2}\right\}$$

Out[40]= $$\left\{u \to \frac{a^2 x}{x^2+y^2}, v \to \frac{a^2 y}{x^2+y^2}\right\}$$

We can recast the above solution into a set of rules that define the individual transistion functions from the black (north projection) to the red (south projection) coordinates. Convert the right hand sides of the above rules to pure functions and then set them as pure function rules for the red coordinate functions.

```
In[41]:= Function /@ Part[solsouth, All, 2] // CoordinatesToTensors[{x, y}] //
           UseCoordinates[{#1, #2}];
         SetTensorValueRules[xu[red@i], %]
         TensorValueRules[x]
```

Out[43]= $$\left\{x^1 \to \left(\frac{a^2\,\#1}{\#1^2+\#2^2}\,\&\right), x^2 \to \left(\frac{a^2\,\#2}{\#1^2+\#2^2}\,\&\right)\right\}$$

Then we can display and evaluate the coordinate transfer functions in normal textbook style.

```
In[44]:= xu[red@a][xu[b]]
         % // EinsteinArray[] // EinsteinArgument[x]
         % /. TensorValueRules[x]
```

Out[44]= $x^a[x^b]$

Out[45]= $\{x^1[x^1, x^2], x^2[x^1, x^2]\}$

Out[46]= $$\left\{\frac{a^2 x^1}{(x^1)^2+(x^2)^2}, \frac{a^2 x^2}{(x^1)^2+(x^2)^2}\right\}$$

We can even calculate the Jacobean of the transistion function using this functional notation.

```
In[47]:= SetAttributes[a, {Constant}]
         xu[red@a][xu[b]]
         PartialD[labs][%, xu[c]]
         % // EinsteinArray[] // EinsteinArgument[x] // KroneckerEvaluate[δ] // MatrixForm
         (step1 = % /. TensorValueRules[x] // Simplify) // MatrixForm
```

Out[48]= $x^a[x^b]$

Out[49]= $\dfrac{\partial x^a[x^b]}{\partial x^c}$

Out[50]//MatrixForm=

$$\begin{pmatrix} x^{1\,(1,0)}[x^1, x^2] & x^{1\,(0,1)}[x^1, x^2] \\ x^{2\,(1,0)}[x^1, x^2] & x^{2\,(0,1)}[x^1, x^2] \end{pmatrix}$$

Out[51]//MatrixForm=

$$\begin{pmatrix} \frac{a^2\left(-(x^1)^2+(x^2)^2\right)}{\left((x^1)^2+(x^2)^2\right)^2} & -\frac{2\,a^2\,x^1\,x^2}{\left((x^1)^2+(x^2)^2\right)^2} \\ -\frac{2\,a^2\,x^1\,x^2}{\left((x^1)^2+(x^2)^2\right)^2} & \frac{a^2\left((x^1)^2-(x^2)^2\right)}{\left((x^1)^2+(x^2)^2\right)^2} \end{pmatrix}$$

A more convenient method is to set to set the values as algebraic expressions...

```
In[52]:= Part[solsouth, All, 2] // CoordinatesToTensors[{x, y}];
         SetTensorValueRules[xu[red@i], %]
         TensorValueRules[x]
```

Out[54]= $\left\{x^1 \to \dfrac{a^2\,x^1}{(x^1)^2+(x^2)^2},\ x^2 \to \dfrac{a^2\,x^2}{(x^1)^2+(x^2)^2}\right\}$

and then calculate more directly.

```
In[55]:= xu[red@a]
         PartialD[labs][%, xu[c]]
         (step2 = % // ToArrayValues[] // KroneckerEvaluate[δ] // Simplify) // MatrixForm
         step2 == step1 // Simplify
```

Out[55]= x^a

Out[56]= $\dfrac{\partial x^a}{\partial x^c}$

Out[57]//MatrixForm=

$$\begin{pmatrix} -\frac{a^2\left((x^1)^2-(x^2)^2\right)}{\left((x^1)^2+(x^2)^2\right)^2} & -\frac{2\,a^2\,x^1\,x^2}{\left((x^1)^2+(x^2)^2\right)^2} \\ -\frac{2\,a^2\,x^1\,x^2}{\left((x^1)^2+(x^2)^2\right)^2} & \frac{a^2\left((x^1)^2-(x^2)^2\right)}{\left((x^1)^2+(x^2)^2\right)^2} \end{pmatrix}$$

Out[58]= **True**

For the polar coordinates, we have that the angle in the equatorial plane is the same and we only have to solve for the relation between the radii, **r** and **s**, of the two projections.

```
In[59]:= transequations =
          Part[polarStereographic, 2] == (Part[polarStereographic, 2] {1, 1, -1} /. {r → s})
         rsols = Solve[transequations, r][[1,1]]
```

Out[59]= $\left\{\frac{2\,a^2\,r\,\mathrm{Cos}[\phi]}{a^2+r^2},\ \frac{2\,a^2\,r\,\mathrm{Sin}[\phi]}{a^2+r^2},\ \frac{-a^3+a\,r^2}{a^2+r^2}\right\} == \left\{\frac{2\,a^2\,s\,\mathrm{Cos}[\phi]}{a^2+s^2},\ \frac{2\,a^2\,s\,\mathrm{Sin}[\phi]}{a^2+s^2},\ -\frac{-a^3+a\,s^2}{a^2+s^2}\right\}$

Out[60]= $r \to \frac{a^2}{s}$

Mathematicians talk about 'gluing together' overlapping coordinate patches. It is essentially the transistion functions that glue them together.

Of course, mathematicians are often blithely unaware of some of the 'engineering' aspects of their mappings. The polar coordinate mapping above was too coarse near the opposite pole and to crowded near the projecting pole. We can't just use greater or fewer points without running into a problem at one end or the other. We can improve the mapping by using **r → Exp[s] - 1** and then using even spacings of **s**.

```
In[61]:= exponentialPolar = Part[polarStereographic, 2] /. r → Exp[s] - 1 // Simplify
```

Out[61]= $\left\{\frac{2\,a^2\,(-1+e^s)\,\mathrm{Cos}[\phi]}{a^2+(-1+e^s)^2},\ \frac{2\,a^2\,(-1+e^s)\,\mathrm{Sin}[\phi]}{a^2+(-1+e^s)^2},\ \frac{a\,(-a^2+(-1+e^s)^2)}{a^2+(-1+e^s)^2}\right\}$

Giving us the patches...

Two Coordinate Patches Covering the Sphere
Exponential Polar Coordinates





North Pole Projection **South Pole Projection**

That gives a more uniform coordinate system.

It is difficult to pull off the same trick with Cartesian coordinates. We can do a little better with something like...

```
In[63]:= powerCartesian =
          Part[cartesianStereographic, 2] /. {x → 4 x + 40 x^3, y → 4 y + 40 y^3} // Simplify
```

Out[63]=

$$\left\{\frac{8\,a^2\,(x+10\,x^3)}{a^2+(4\,x+40\,x^3)^2+(4\,y+40\,y^3)^2},\ \frac{8\,a^2\,(y+10\,y^3)}{a^2+(4\,x+40\,x^3)^2+(4\,y+40\,y^3)^2},\ \frac{a\left(-a^2+(4\,x+40\,x^3)^2+(4\,y+40\,y^3)^2\right)}{a^2+(4\,x+40\,x^3)^2+(4\,y+40\,y^3)^2}\right\}$$

Giving us the patches...



■ 2.3 Example - The Projective Plane

The projective plane is an example of a manifold that is not simply a surface embedded in a higher dimensional space. The elements of the projective plane, denoted as $\mathbf{PR}^2$, are all the lines through the origin in 3-dimensional space as illustrated in the following graphic.



Unfortunately, this graphic is deficient in a number of ways. Only a random sample of 50 lines is shown. In actual fact there are an infinite number of lines and they would completely fill the 3D space. The lines were given random hues but this was only to make them easier to visualize if the lines were to be rotated in an animation. (This does suggest that we could use color to indicate a scalar function on the projective plane.) Finally, the lines would actually be infinite in length but here they are terminated by the edges of the bounding box.

Since each line has a direction in space we could think of identifying each line with a point on the unit sphere. However the lines do not have an orientation and each line intersects a unit sphere at two anti-podal points. To accurately represent the projective plane we must identify those two points as being the same point. So that is another definition of the projective plane, a unit sphere with anti-podal points identified.

The following cell illustrates a patch on $\mathbf{PR}^2$.



The coordinate patch consists of all the lines that intersect the open red disk on the unit sphere. Of course, they also intersect the open anti-podal disk. If we tried to increase the size of the disk to include more than a hemisphere then it would overlap the anti-podal disk and we would not have a 1-1 coordinate mapping function. Two points in the disk would map to the same line.

A convenient method for putting a coordinate patch on a projective plane is to place a tangent plane to the unit sphere and find the location in the plane where a given line intersects the plane. Let's make a 'canonical' algebraic representation of a line. For the canonical representation we can use the lexicographically positive point, `P == {X, Y, Z}`, on the unit sphere that intersects the line. (The negative then gives the anti-podal point.) Let's take the first tangent plane to be the plane with `Z == 1`. The following diagram shows the sectional cut through the unit-sphere that contains the point `P` and its intersection with the tangent plane.



Let **r** be the distance of the intersection point from the origin of the tangent plane. Let **R** be the distance of **P** from the Z-axis. These quantities are related by...

```
In[69]:=  (eqns = {R^2 + Z^2 == 1, Z == R / r}) // TableForm
          RZsol = Solve[eqns, {R, Z}][[2]]
```

Out[69]//TableForm=

$$R^2 + Z^2 == 1$$
$$Z == \frac{R}{r}$$

Out[70]= $\left\{R \to \frac{r}{\sqrt{1+r^2}},\ Z \to \frac{1}{\sqrt{1+r^2}}\right\}$

We just have to plug through the calculations for three coordinate patches. I will use the coordinates **{x1, y1}** for the first patch, **{x2, z2}** for the second patch, and **{y3, z3}** for the third patch. I will use **Φ[i]** for the i'th coordinate map and **ΦI[i]** for the inverse map.The inverse coordinate map to the tangent plane in polar and Cartesian coordinates for the first patch is then...

```
In[71]:= {X, Y, Z} == (Flatten[{R {Cos[ϕ], Sin[ϕ]}, Z}] /. RZsol)
         step1 = % /. Thread[r {Cos[ϕ], Sin[ϕ]} → {x1, y1}] /. r^2 → x1^2 + y1^2
         ΦI[1][x1_, y1_] = Part[%, 2]
```

Out[71]= $\{X, Y, Z\} == \left\{\frac{r\,\mathrm{Cos}[\phi]}{\sqrt{1+r^2}}, \frac{r\,\mathrm{Sin}[\phi]}{\sqrt{1+r^2}}, \frac{1}{\sqrt{1+r^2}}\right\}$

Out[72]= $\{X, Y, Z\} == \left\{\frac{x1}{\sqrt{1+x1^2+y1^2}}, \frac{y1}{\sqrt{1+x1^2+y1^2}}, \frac{1}{\sqrt{1+x1^2+y1^2}}\right\}$

Out[73]= $\left\{\frac{x1}{\sqrt{1+x1^2+y1^2}}, \frac{y1}{\sqrt{1+x1^2+y1^2}}, \frac{1}{\sqrt{1+x1^2+y1^2}}\right\}$

The coordinate map itself is given by...

```
In[74]:= Join[step1 // Thread, {X^2 + Y^2 + Z^2 == 1}]
         Solve[%, {x1, y1}][[1]]
         Φ[1][X_, Y_, Z_] = {x1, y1} /. %
```

Out[74]= $\left\{X == \frac{x1}{\sqrt{1+x1^2+y1^2}}, Y == \frac{y1}{\sqrt{1+x1^2+y1^2}}, Z == \frac{1}{\sqrt{1+x1^2+y1^2}}, X^2+Y^2+Z^2 == 1\right\}$

Out[75]= $\left\{x1 \to \frac{X}{Z}, y1 \to \frac{Y}{Z}\right\}$

Out[76]= $\left\{\frac{X}{Z}, \frac{Y}{Z}\right\}$

This coordinate map covers almost the entire projective plane. The only lines that are missing are those that lie in the equatorial plane. We will need two additional orthogonal coordinate planes to cover those lines. We can put the second tangent plane at **Y == 1**.

```
In[77]:= {X, Y, Z} == ({R Sin[ϕ], Y, R Cos[ϕ]} /. (RZsol /. Z → Y))
         step2 = % /. Thread[r {Sin[ϕ], Cos[ϕ]} → {x2, z2}] /. r^2 → x2^2 + z2^2
         ΦI[2][x2_, z2_] = Part[%, 2]
```

Out[77]= $\{X, Y, Z\} == \left\{\frac{r\,\mathrm{Sin}[\phi]}{\sqrt{1+r^2}}, \frac{1}{\sqrt{1+r^2}}, \frac{r\,\mathrm{Cos}[\phi]}{\sqrt{1+r^2}}\right\}$

Out[78]= $\{X, Y, Z\} == \left\{\frac{x2}{\sqrt{1+x2^2+z2^2}}, \frac{1}{\sqrt{1+x2^2+z2^2}}, \frac{z2}{\sqrt{1+x2^2+z2^2}}\right\}$

Out[79]= $\left\{\frac{x2}{\sqrt{1+x2^2+z2^2}}, \frac{1}{\sqrt{1+x2^2+z2^2}}, \frac{z2}{\sqrt{1+x2^2+z2^2}}\right\}$

```
In[80]:= Join[step2 // Thread, {X^2 + Y^2 + Z^2 == 1}]
         Solve[%, {x2, z2}][[1]]
         Φ[2][X_, Y_, Z_] = {x2, z2} /. %
```

Out[80]= $\left\{X == \frac{x2}{\sqrt{1 + x2^2 + z2^2}}, Y == \frac{1}{\sqrt{1 + x2^2 + z2^2}}, Z == \frac{z2}{\sqrt{1 + x2^2 + z2^2}}, X^2 + Y^2 + Z^2 == 1\right\}$

Out[81]= $\left\{x2 \to \frac{X}{Y}, z2 \to \frac{Z}{Y}\right\}$

Out[82]= $\left\{\frac{X}{Y}, \frac{Z}{Y}\right\}$

That covers all of the equatorial lines except the one that lies along the X-axis. We can put the third tangent at **X == 1** to capture that line.

```
In[83]:= {X, Y, Z} == ({X, R Cos[ϕ], R Sin[ϕ]} /. (RZsol /. Z → X))
         step3 = % /. Thread[r {Cos[ϕ], Sin[ϕ]} → {y3, z3}] /. r^2 → y3^2 + z3^2
         ΦI[3][y3_, z3_] = Part[%, 2]
```

Out[83]= $\{X, Y, Z\} == \left\{\frac{1}{\sqrt{1 + r^2}}, \frac{r\,\mathrm{Cos}[\phi]}{\sqrt{1 + r^2}}, \frac{r\,\mathrm{Sin}[\phi]}{\sqrt{1 + r^2}}\right\}$

Out[84]= $\{X, Y, Z\} == \left\{\frac{1}{\sqrt{1 + y3^2 + z3^2}}, \frac{y3}{\sqrt{1 + y3^2 + z3^2}}, \frac{z3}{\sqrt{1 + y3^2 + z3^2}}\right\}$

Out[85]= $\left\{\frac{1}{\sqrt{1 + y3^2 + z3^2}}, \frac{y3}{\sqrt{1 + y3^2 + z3^2}}, \frac{z3}{\sqrt{1 + y3^2 + z3^2}}\right\}$

The coordinate mapping solution with *Mathematica* is a little irritating because we have to use a substitution rule.

```
In[86]:= Join[step3 // Thread, {X^2 + Y^2 + Z^2 == 1}]
         Solve[%, {y3, z3}][[1]]
         % /. Y^2 + Z^2 - 1 → -X^2
         Φ[3][X_, Y_, Z_] = {y3, z3} /. %
```

Out[86]= $\left\{X == \frac{1}{\sqrt{1 + y3^2 + z3^2}}, Y == \frac{y3}{\sqrt{1 + y3^2 + z3^2}}, Z == \frac{z3}{\sqrt{1 + y3^2 + z3^2}}, X^2 + Y^2 + Z^2 == 1\right\}$

Out[87]= $\left\{y3 \to -\frac{X\,Y}{-1 + Y^2 + Z^2}, z3 \to -\frac{X\,Z}{-1 + Y^2 + Z^2}\right\}$

Out[88]= $\left\{y3 \to \frac{Y}{X}, z3 \to \frac{Z}{X}\right\}$

Out[89]= $\left\{\frac{Y}{X}, \frac{Z}{X}\right\}$

That gives us all three coordinate mappings and their inverses and completely covers the lines in the projective plane. Next we have to solve for the transistion functions. I will denote the transistion functions by `ΦT[from chart, to chart][from coordinates]`. All the transistion functions are...

```
In[90]:= ΦT[1, 2][x1_, y1_] = Φ[2] @@ ΦI[1][x1, y1]
```

Out[90]= $\left\{\frac{x1}{y1}, \frac{1}{y1}\right\}$

```
In[91]:= ΦT[1, 3][x1_, y1_] = Φ[3] @@ ΦI[1][x1, y1]
```

Out[91]= $\left\{\frac{y1}{x1}, \frac{1}{x1}\right\}$

```
In[92]:= ΦT[2, 3][x2_, z2_] = Φ[3] @@ ΦI[2][x2, z2]
```

Out[92]= $\left\{\frac{1}{x2}, \frac{z2}{x2}\right\}$

```
In[93]:= ΦT[2, 1][x2_, z2_] = Φ[1] @@ ΦI[2][x2, z2]
```

Out[93]= $\left\{\frac{x2}{z2}, \frac{1}{z2}\right\}$

```
In[94]:= ΦT[3, 1][y3_, z3_] = Φ[1] @@ ΦI[3][y3, z3]
```

Out[94]= $\left\{\frac{1}{z3}, \frac{y3}{z3}\right\}$

```
In[95]:= ΦT[3, 2][y3_, z3_] = Φ[2] @@ ΦI[3][y3, z3]
```

Out[95]= $\left\{\frac{1}{y3}, \frac{z3}{y3}\right\}$

These apply, of course, only in the region of overlap. The following graphic illustrates the `ΦI[1]` inverse coordinate map for a limited portion of the x1-y1 plane.

Φ[1] Chart on Projective Plane



Chart 1



First Coordinate Patch

The small squares in the chart were mapped into their corresponding polygons, which roughly cover the northern hemisphere of the unit-sphere. The projective plane lines were taken through the center of each polygon. These charts are not highly 'engineered' in the sense that they do not give a very uniform distribution of the projective plane lines.

3. A Scalar Field on a Manifold

We'll use Example 2.2 in Section 2 to illustrate how we represent a scalar field on a manifold. We will show that the functions may be quite different on two charts but that they still represent the same function on the manifold. This section also illustrates the technique for transfering a contour plot from a chart to the manifold. We will use the exponential polar coordinates introduced in Section 2.2. Our two stereographic projections for a sphere of radius 1 are...

In[97]:= $\texttt{expnorthproj}[r\_, \phi\_] = \texttt{Simplify}\left[\dfrac{a\,\{2\,a\,r\,\texttt{Cos}[\phi],\ 2\,a\,r\,\texttt{Sin}[\phi],\ r^2 - a^2\}}{a^2 + r^2} \;/.\; a \to 1 \;/.\; r \to e^r - 1\right]$

$\texttt{expsouthproj}[r\_, \phi\_] = \texttt{Simplify}\left[\dfrac{a\,\{2\,a\,r\,\texttt{Cos}[\phi],\ 2\,a\,r\,\texttt{Sin}[\phi],\ -(r^2 - a^2)\}}{a^2 + r^2} \;/.\; a \to 1 \;/.\; r \to e^r - 1\right]$

Out[97]= $\left\{\dfrac{2\,(-1 + e^r)\,\texttt{Cos}[\phi]}{1 + (-1 + e^r)^2},\ \dfrac{2\,(-1 + e^r)\,\texttt{Sin}[\phi]}{1 + (-1 + e^r)^2},\ \dfrac{e^r\,(-2 + e^r)}{1 + (-1 + e^r)^2}\right\}$

Out[98]= $\left\{\dfrac{2\,(-1 + e^r)\,\texttt{Cos}[\phi]}{1 + (-1 + e^r)^2},\ \dfrac{2\,(-1 + e^r)\,\texttt{Sin}[\phi]}{1 + (-1 + e^r)^2},\ -\dfrac{e^r\,(-2 + e^r)}{1 + (-1 + e^r)^2}\right\}$

The following function, **`fnorth`**, will be our function in the equatorial plane for the north projection. (I simply made it up as an example.) We also convert it to the function **`gnorth`**, which is in terms of our exponential radial coordinate.

In[99]:= $\texttt{fnorth}[r\_, \phi\_] = 10\,e^{-r/2}\,\texttt{BesselJ}[3,\ 4\,r]\,\texttt{Cos}[\phi]$

$\texttt{gnorth}[r\_, \phi\_] = \texttt{fnorth}[r, \phi] \;/.\; r \to \texttt{Exp}[r] - 1 \;//\; \texttt{Simplify}$

Out[99]= $10\,e^{-r/2}\,\texttt{BesselJ}[3,\ 4\,r]\,\texttt{Cos}[\phi]$

Out[100]=

$10\,e^{\frac{1}{2} - \frac{e^r}{2}}\,\texttt{BesselJ}[3,\ 4\,(-1 + e^r)]\,\texttt{Cos}[\phi]$

We can obtain the corresponding functions for the south projection by using the transition function **$r \to 1/r$** (obtained in Section 2.2) and then making the exponential substitution.

In[101]:=

$\texttt{fsouth}[r\_, \phi\_] = \texttt{fnorth}[r, \phi] \;/.\; r \to 1/r \;//\; \texttt{Simplify}$

$\texttt{gsouth}[r\_, \phi\_] = \texttt{fsouth}[r, \phi] \;/.\; r \to \texttt{Exp}[r] - 1 \;//\; \texttt{Simplify}$

Out[101]=

$10\,e^{-\frac{1}{2r}}\,\texttt{BesselJ}\left[3,\ \dfrac{4}{r}\right]\,\texttt{Cos}[\phi]$

Out[102]=

$10\,e^{\frac{1}{2-2e^r}}\,\texttt{BesselJ}\left[3,\ \dfrac{4}{-1 + e^r}\right]\,\texttt{Cos}[\phi]$

The north and south functions are clearly different functions as we can see by taking a radial slice at zero angle...

In[103]:=

```
Plot[{gnorth[r, 0], gsouth[r, 0]}, {r, 10^-8, 1.09861}];
```



For the patch size that corresponds to that used in Section 2.2 we use the following exponential **r** maximum.

In[104]:=

```
FindRoot[Exp[r] - 1 == 2, {r, 2}]
```

Out[104]=

```
{r → 1.09861}
```

The maximum and minimum values of our function are.

In[105]:=

```
NMaximize[{gnorth[r, φ], 0 < r < 1.098, π < φ < 3 π}, {r, φ}]
NMinimize[{gnorth[r, φ], 0 < r < 1.098, 0 < φ < π}, {r, φ}]
```

Out[105]=

```
{2.61138, {r → 0.685516, φ → 6.28319}}
```

Out[106]=

```
{-2.61138, {r → 0.685516, φ → 3.14159}}
```

The following two plots display the function under it mapping to the two charts.

gnorth[r, ϕ]



gsouth[r, ϕ]



The following is a contour plot of the **`gnorth`** in the exponential **`r`**, ϕ coordinates.



That is our actual mapping domain, but for convenience we can display this in the equatorial plane.



Looking at the function for the southern projection we obtain the clearly different picture...



We cannot directly use these contour regions to map to the sphere. The colored regions are like solid blocks that won't bend to the surface. Instead we have to make a grid of polygons and then trim it to fit the various contour regions. This is done with the routines **`MakePolyGrid`** and **`TrimPolygons`** from the **`DrawGraphics`** package.

Making and trimming polygonal grid for mapping to the sphere...

In[115]:=
```
polygrid = MakePolyGrid[{30, 40}, {{10^-8, 0}, {1.09861, 2 π}}] // N;
```

The following takes 45 seconds on an 800MH machine, and the second one takes about 63 seconds.

In[116]:=
```
Do[polysnorth[i] = polygrid // TrimPolygons[gnorth, {contours⟦i⟧, contours⟦i + 1⟧}],
  {i, 1, Length[contours] - 1}] // Timing
```

Out[116]=
```
{43.61 Second, Null}
```

In[117]:=
```
Do[polyssouth[i] = polygrid // TrimPolygons[gsouth, {contours⟦i⟧, contours⟦i + 1⟧}],
  {i, 1, Length[contours] - 1}] // Timing
```

Out[117]=
```
{61.13 Second, Null}
```

The following displays all the trimmed polygonial grids for the various contour regions. Again, for convenience, the rectangular grid has been transformed to the circular region in the equatorial plane of the projection.

In[118]:=

```
Draw2D[
 {Table[{colorfun[(contours[[i+1]] + contours[[i]]) / 2], polysnorth[i],
     polysnorth[i] // PolygonOutline[GrayLevel[0.7]]}, {i, 1, Length[contours] - 1}]} /.
  DrawingTransform[#1 Cos[#2] &, #1 Sin[#2] &],
 AspectRatio → Automatic,
 Frame → True,
 PlotLabel → HoldForm[gnorth[r, ϕ]],
 PlotRange → Automatic,
 TextStyle → {FontSize → 12, FontWeight → "Bold", FontFamily → "Times"},
 ImageSize → 300];
```



```
In[119]:=
    Draw2D[
      {Table[{colorfun[(contours[[i+1]] + contours[[i]]) / 2], polyssouth[i],
          polyssouth[i] // PolygonOutline[GrayLevel[0.7]]}, {i, 1, Length[contours] - 1}]} /.
       DrawingTransform[#1 Cos[#2] &, #1 Sin[#2] &],
      AspectRatio → Automatic,
      Frame → True,
      PlotLabel → HoldForm[gsouth[r, ϕ]],
      PlotRange → Automatic,
      TextStyle → {FontSize → 12, FontWeight → "Bold", FontFamily → "Times"},
      ImageSize → 300];
```



Finally, the graphic from the following closed cell projects the two functions onto the sphere showing that they give the same scalar field in the region of overlap.

Scalar Field Represented by Two Coordinate Patches
Exponential Polar Coordinates



North Pole Projection **South Pole Projection**

There is a single unique scalar field on the manifold. But we have to represent the field in the coordinate domains for the various charts and their functional form in these different charts is different. But there are transformations from one to the other.

Bibliography

Bishop, Richard L. & Goldberg, Samuel I. (1968). *Tensor Analysis on Manifolds*. Dover Publications, New York.

Foster, J. & Nightingale, J.D. (1995). *A Short Course in General Relativity, Second Edition*. Springer, New York.

Gray, Alfred (1998). *Modern Differential Geometry of Curves and Surfaces with Mathematica*, Second Edition. CRC Press, New York.

Halmos, Paul R. (1974). *Finite-Dimensional Vector Spaces*, Springer-Verlag, New York

Hartle, James B. (2003). *Gravity: An Introduction to Einstein's General Relativity*. Addison Wesley, San Francisco.

Lovelock, David & Rund, Hanno. (1975). *Tensors, Differential Forms, and Variational Principles*. Dover Publications, New York.

Ludvigsen, Malcolm. (1999). *General Relativity, A Geometric Approach*. Cambridge University Press, Cambridge.

Charles W. Misner, Kip S. Thorne & John Archibald Wheeler. (1973), *Gravitation*, W.H. Freeman and Company, San Francisco

O'Neill, Barrett (1966). *Elementary Differential Geometry*. Academic Press, New York.

Penrose, Roger (2005). *The Road to Reality*. Alfred A. Knopf, New York.

Schutz, Bernard F. (1980). *Geometrical Methods of Mathematical Physics*. Cambridge University Press, Cambridge.

Schutz, Bernard F. (1990). *A First Course in General Relativity*. Cambridge University Press, Cambridge.

Spivak, Michael. (1970). *A Comprehensive Introduction to Differential Geometry*. Publish or Parish, Boston

Wald, Robert M. (1984). *General Relativity*. The University of Chicago Press, Chicago.